
Abstract

The Resource-Constrained Project Scheduling Problem (RCPSP) is a long-standing challenge problem in the area
of constraint optimization. Its goal is to schedule partially ordered activities, each requiring specific resources,
as effectively as possible while respecting the resources’ capacity limits. Previous approaches have utilized
Constraint Programming or local search techniques for solving the RCPSP to exact or approximate optimality,
respectively. While Answer Set Programming (ASP) along with its extensions by difference logic or finite domain
constraints has been successfully applied to closely related scheduling problems, these declarative problem solving
approaches have so far not been explored for the RCPSP.

In this work, we elaborate on modeling approaches for encoding the RCPSP in plain ASP as well as its
extensions by difference logic and finite domain constraints. Interestingly, formulating resource capacity limits is
simpler in plain ASP due to the explicit representation of time points, while the use of difference logic or finite
domain constraints incurs a certain modeling effort. Our empirical comparison on small- to medium-scale RCPSP
instances shows that the simplicity of modeling promotes plain ASP and more than outweighs its larger problem
representation size, which counteracts the performance trends typically observed on related scheduling problems.

1. Introduction

The Resource-Constrained Project Scheduling Problem (RCPSP) [1] is a special class of scheduling
problems, particularly interesting due to its various applications. Some of the famous use cases include
experiments scheduling on the Rosetta/Philae mission [2], scheduling the Airbus A330 assembly line 3],
scheduling hazardous material examinations [4], and many others. The idea is to find a schedule of
the set of activities within a project while considering limited resources. Each activity takes a certain
amount of time and resources. Additionally, activities are subject to precedence constraints, meaning
that some activities must finish before others can start. The objective is to minimize the overall project
duration while adhering to the resource and precedence constraints. There are several variants of the
RCPSP, including cases with stochastic activity durations, renewable and non-renewable resources,
and time-dependent resource availability. In this paper, we consider the deterministic RCPSP, where
activity durations are known in advance and resources are renewable (i.e., not permanently consumed
by a process) with fixed availability. Furthermore, the problem parameters are instance-dependent, i.e.,
different instances may involve varying numbers of activities, resources, and precedence relations.
The RCPSP was introduced, formalized, and studied as part of operations research dating back to the
1960s (see early surveys [5, 6]), and also serves as the backbone of planning and execution in project
management [7, 8, 9]. One can view the RCPSP as an evolution of the project scheduling problem
without resource constraints, which was efficiently solved with the Critical Path Method (CPM) [10].
Other tools, such as the Program Evaluation and Review Technique (PERT), were focused on analyzing

scheduling problems with nondeterministic activity durations. However, these tools analyze project
networks solely based on the precedence constraints [10, 11] and do not explicitly account for resource
limitations, thus motivating the need for more sophisticated formulations [12, 13]. The complexity of
RCPSP was established as NP-hard [12] when resources are not fixed.

In the early days, the main approaches to RCPSP solving were Integer Programming [14] and Branch-
and-Bound methods [15]. However, due to the problem’s complexity, these methods could only solve
small instances in manageable time. This led to continuing the research in the direction of heuristics
(see [16] for an overview), aiming to provide good-enough solutions without guaranteeing optimality.
Moreover, these methods were improved with metaheuristic approaches, such as genetic algorithms,
simulated annealing, and tabu search (see [17] for a survey), which were capable of escaping local
optima and handling larger instances, but still without providing optimality guarantees. With the
beginning of this millennium and advancements of exact solving methods, the RCPSP received renewed
interest from Constraint Programming (CP) [18] and Mixed Integer Linear Programming (MILP) [19]
communities. Furthermore, these approaches were combined with metaheuristics, such as Large
Neighborhood Search (LNS) [3, 20], resulting in significant improvements in solution quality. However,
these hybrid methodologies often require expert knowledge of problem-solving tools and problem
specifics for successful implementation. Recent work on the RCPSP has explored machine-learning-
guided heuristics and reinforcement learning methods to address uncertainty and dynamic scheduling
environments [21, 22]. All these efforts resulted in tools capable of solving large industry-scale instances.

So far, CP and MILP are the main exact solving approaches for the RCPSP. Our work explores
Answer Set Programming (ASP) [23] as a high-level declarative programming language, which has
been successfully applied to related scheduling problems (see, e.g., [24]), but not yet to the RCPSP.
As a language oriented primarily towards NP-hard search problems, ASP is naturally suited for this
task. Aside from modeling the RCPSP in plain ASP, we will investigate two ASP extensions: Difference
Logic (DL) and finite domain constraints. DL aims to improve the modeling and solving of problems
with numerical constraints by incorporating a restricted form of linear arithmetics in ASP modulo
DL [25]. Similarly, Constraint Answer Set Programming (CASP) [26] integrates finite domain constraints
over bounded integers. CASP solving approaches, including specializations to ASP modulo DL, can
be grouped into three categories: hybrid systems that combine an ASP solver with a CP or MILP
solver encompass acsolver [27], ascass [28], caspscript [29], clingcon v1 [30] and v2 [31], clingo[LP] [25],
divhex[CP] [32], and ezcsp [33]; translations mapping CASP to MILP or Satisfiability Modulo Theories
(SMT) [34] formulations are performed by aspmt2smt [35], dingo [36], ezsmt [37], and mingo [38]; the
systems adsolver [39], clingcon v3 [40], clingo[DL] [25], idp [41], and inca [42] implement extensions of
ASP systems for constraint propagation on theory atoms. Among the latter, clingcon, clingo[DL], and
inca augment the ASP system clingo [43] by performing specific forms of lazy clause generation [44], a
technique from satisfiability solving driving the success of the state-of-the-art CP solver OR-Tools [45].

Since the RCPSP is a temporal scheduling problem, any declarative representation, including ASP
encodings, must explicitly or implicitly represent time. In plain ASP, the time is represented explicitly as
a finite subset of natural numbers. Due to expressivity gains of DL and finite domain constraints, the time
values can be left implicit in these ASP extensions. The chosen representation of time influences how
the problem constraints are formulated. Although an implicit time representation is generally preferable
from both modeling and computational perspectives, in this work we observe the opposite when
modeling the RCPSP in ASP. Specifically, resource constraints are more convenient to express using an
explicit time representation. This is because the resource capacity limits apply at each point in time, and
extracting specific events of interest to substitute explicit time points incurs some extra modeling effort
for ASP modulo DL and CASP formulations. Such inherent differences in the modeling approaches
impact the computational performance, where our empirical comparison on small- to medium-scale
RCPSP instances yields that plain ASP’s simplicity outweighs its larger problem representation size,
which comes as a somewhat surprising difference to usual performance trends on temporal scheduling
problems [46, 47].

The main contributions of this work are twofold:



« We propose encodings for the RCPSP in ASP, ASP modulo DL, and CASP, and analyze them from
a modeling perspective.

« We empirically compare our RCPSP modeling approaches on small- to medium-scale RCPSP
instances [48], where we incorporate the state-of-the-art CP solver OR-Tools as a baseline.

The remainder of the paper is structured as follows. We start by providing preliminaries for this
work in Section 2. Section 3 elaborates our plain ASP, ASP modulo DL, and CASP formulations of the
RCPSP. In Section 4, we present our experimental setup and results. Finally, Section 5 concludes the
paper by discussing the main findings and outlining future work.

2. Preliminaries

We first provide a brief overview of ASP, ASP modulo DL, and CASP from a problem modeling perspec-
tive, following the presentation of [47]. Second, we introduce and illustrate the RCPSP.

2.1. Answer Set Programming

ASP [23] is a knowledge representation and reasoning paradigm geared towards the modeling and
solving of combinatorial (optimization) problems. A (first-order) ASP program consists of rules of the
formh:- by,...,by,., in which the head h is an atom p(t1,...,t,;) or a choice {p(t1,...,t;,)}, and
each body literal b; is an atom p(t1,...,tm), possibly preceded by the default negation connective not
and/or followed by a condition : cq,...,cy, abuilt-in comparison t] o tg witho € {< <=, = 1= 5= >},
or an aggregate #sum{ti,...,ty, : c1,...,¢} O tyq1. Each t; denotesaterm, i.e., a constant, variable,
tuple, or arithmetic expression, and each element cj of a condition is an atom that may be preceded
by not or a built-in comparison. The head h or body by,. .. ,b, may be omitted, in which case we call
the rule an integrity constraint or fact, respectively. Roughly speaking, an ASP program is a shorthand
for its ground instantiation, obtainable by substituting variables with all of the available constants and
evaluating arithmetic expressions, and the semantics is given by answer sets, i.e., sets of (true) ground
atoms such that all rules of the ground instantiation are satisfied and allow for deriving each of the
ground atoms by some rule whose body is satisfied. The ground instantiation process is detailed in [49],
and the answer set semantics is further elaborated in [50, 23].

ASP modulo DL integrates DL constraints [51], i.e., expressions written as &piFr{t] - to} <= t3, in
the heads and bodies of rules. Except for integers, which denote their numerical values, the terms
t1 and to represent DL variables that can be assigned any integer value. However, the difference
t] - to must not exceed the integer t3 for the DL constraint to be satisfied. That is, the evaluation of
DL constraints interacts with the encompassing rules, and the clingo[DL] system [25] extends clingo
by assuring the consistency of DL constraints imposed by an answer set. If these DL constraints are
satisfiable, a canonical assignment of smallest feasible integer values to DL variables can be determined
in polynomial time and is output along with the answer set by clingo[DL].

Finally, clingcon [31] extends the ASP syntax with expressions of the form &sum{ly*z1; lo*zo;

.3 lp*x,} <= k to represent linear constraints l1z1 + loze + ... + [z, < k, where each [; is an
integer, each z; is a variable, and k is an arithmetic expression over integers and variables. Moreover,
atoms of the form &pom{l; . .ls} = x allow for limiting the domain of a variable x to the range [I1, [2].
Different from clingo[DL], it is also possible to include &minimize statements referring to the sum of
the contained variables’ values. The main differences between clingo[DL] and clingcon are outlined in
Table 1. While clingcon’s linear constraints permit modeling complex problems, the consistency of DL
constraints can be decided in polynomial time by clingo[DL]. Moreover, all theory atoms of clingcon
are external and do thus not need any justification to be relevant to an answer set, while clingo[DL]
associates this meaning to theory atoms in rule bodies only.



Table 1
Overview of clingo[DL] and clingcon concepts

Concept clingo[DL] clingcon

Constraint type r—y<k lhixi+- -+ o, <k
Theory complexity ~ Polynomial-time decidable NP-complete

Theory atoms Head atoms defined; body atoms external  All theory atoms are external
Syntax &DIFF{x -y} <= 5 &suM{3*x; 4%y} <= z-7

2.2. Resource-Constrained Project Scheduling Problem (RCPSP)

The RCPSP is a canonical combinatorial optimization problem that formally captures the challenges
of scheduling under both precedence and resource constraints [12, 9]. It is known to be NP-hard
in the strong sense [12, 52] and has been extensively studied in the operations research and project
management literature.

Formally, the RCPSP is defined on a project network represented as a directed acyclic graph G =
(V, &), where V = {1,...,n} is a set of activities and £ is a set of precedence relations. The problem is
further characterized by the following concepts:

Activities. Each activity ¢« € )V has an associated duration d; > 0 and must be executed without
interruption once started — a condition known as the non-preemption assumption.

Precedence Constraints. For each arc (4, j) € &, activity j cannot begin before activity ¢ has been
completed. Denoting the start times of the activities by S; and S}, this is expressed as:

Sj > S; +d;, V(’i,j) cé&

Renewable Resources. The project involves a set R = {1,..., m} of renewable resources, where
each resource k£ € R has a fixed capacity Ry, available at every time period. Each activity ¢ € V
requires 7 > 0 units of resource k over its entire duration. At any time point ¢, the total resource
consumption by all activities in execution must not exceed the available capacity:

> riw <Ry, VEER, V
i€A(t)

where A(t) ={i € V| S; <t < S; + d;} denotes the set of activities being executed at time ¢.

Objective. The standard objective of the RCPSP is to find a feasible schedule — that is, an assignment of
start times {S; }ieyp to all activities — such that both the precedence constraints and the resource
constraints are satisfied, while minimizing the project makespan Chyax = max{S; +d; | i € V},
i.e., the latest completion time of any activity.

Example 1. Consider a project consisting of five activities, labeled A, B, C, D, E, and two resources,
blue and green, both having a capacity of 2 units. Activities A and C precede D, and B precedes C
and E. Activities B and D require the green resource in quantities 1 or 2, respectively, activities C' and
E require the blue resource in quantity 1, and activity A requires 1 unit of each resource. Activity D
has a duration of 1 time unit, activities B and C have a duration of 2, and A and F have a duration of 3.

The project network is often visually depicted as a dependency graph, as presented in Figure 1(a). An
optimal schedule for this project is shown in Figure 1(b). Here, the x-axis displays time. Activities are
represented by gray rectangles including their labels, and the colored rectangles below them indicate
resource requirements in terms of the resources’ colors and units. Vertical activity stacking is purely
for visualization, where activities that overlap at time points are executed simultaneously.

This example is designed to point out the following properties that will turn out to be important
from the aspect of modeling:

Figure 1: An example RCPSP instance
task(a,3). res(a,r1l,1). res(a,r2,1). prec(a,d). capacity(r1,2).
task(b,2). res(b,r1,1). prec(b,c). prec(b,e). capacity(r2,2).
task(c,2). res(c,r2,1). prec(c,d).
task(d,1). res(d,r1,2).
task(e,3). res(e,r2,1).

Listing 1: Facts representation of the RCPSP instance in Example 1

+ Merely accumulating the resource requirements of activities that overlap at some time is insuffi-
cient to determine resource consumption. For example, activities A and C overlap, and likewise C'
and F, for a total of 3 activities that require the blue resource. However, only two of the activities
are executed simultaneously. This means that the maximum use of the blue resource is 2, not 3.

« Regarding the availability of resources, activity D could start earlier, i.e., at time point 3 rather
than 4. However, the precedence constraints necessitate that D is executed only after the
completion of C. Conversely, activity F could start earlier from the perspective of its precedence
relation with B, but due the blue resource limit, it needs to wait for 1 resource unit to get free.

« Figure 1(b) displays one of the optimal schedules. An alternative optimal schedule can be obtained
by delaying the execution of activities C' and D by 1 time unit each. In total, our example RCPSP
instance has 5 optimal schedules with a project makespan of 6.

3. Modeling

As customary in ASP, our RCPSP models combine instance-specific facts with a general encoding part.
We start by describing the fact format, then present our plain ASP encoding, and finally turn to ASP
modulo DL and CASP encodings circumventing the explicit representation of time points.

3.1. Fact Format

The format of facts specifying an RCPSP instance, such as the one described in Example 1, is illustrated
by Listing 1. Activities together with their durations are given by the task/2 predicate. The res/3
predicate specifies the activities’ resource requirements, where r1 stands for the green and r2 for
the blue resource according to the coloring scheme of Figure 1(a). Precedence relations between the
activities are expressed by the prec/2 predicate, and the maximum capacity of 2 units for both resources,
r1 and r2, is given by the capacity/2 predicate.

3.2. Plain ASP Encoding

To begin with, the rules in Listing 2 define domain predicates that are deterministically evaluated in
the grounding phase. This includes task/1 to access activities’ labels whenever their durations do not
matter (line 1), first/1 for activities that have no predecessors in the precedence relations (line 2), and

% all activities and activities without any predecessors or successors, respectively
task(I) :-task(I,D).

first(I) :- task(I), not prec(_,I).

final(I) :- task(I), nofl prec(I,_ ).

% reflexive transitive closure of precedence among activities
trans(I1,11) :- task(I1).
trans(I1,I3) :- task(Il), prec(I1,I2), trans(I2,I3).

% activities I1 and I2 without precedence may use a common resource R simultaneously
check(11,I2,R) :-res(I1,R,X1), res(I2,R,X2), capacity(R,C),

not trans(11,12), not trans(I2,I1).
check(I1,12) :-check(I1,I2,R).

% sum of activity durations provides upper bound on makespan and variable domains
horizon(H) :- #sum{D, I : task(I,D)} = H.
range(0..H-1) : - horizon(H).

Listing 2: Domain predicates used by RCPSP encodings

% undone(I,T) expresses that activity I is not yet completed at time T
undone(I,T) :-task(I,D), range(T), T < D.
{undone(I,T)} :-undone(I,T-1), range(T).

% successor activity J takes at least its duration D from completing predecessor I

:-undone(I,T), prec(I,J), task(J,D), T’ = T+D, range(T’-1), nof undone(J,T’).

% resource capacities must not be exceeded at any time T

: - capacity(R,C), range(T),

#sum{X,I : res(I,R,X), task(I,D), undone(I,T), not undone(I,T+D)} > C.

% the times T at which activities without successors aren’t completed yield makespan
:~ final(I), undone(I,T). [1,T]

Listing 3: Plain ASP encoding of RCPSP

similarly final/1 for activities without successors (line 3). The trans/2 predicate, defined by the rules
in lines 7 and 8, provides the reflexive transitive closure of the precedence relations. Note that neither
trans(i,j) nor trans(j,?) holds for (distinct) activities 7, j € V that may be executed simultaneously.
If such activities have a required resource in common, it is indicated by the check/3 predicate (lines
11-12), and check/2 is derived by projection omitting the shared resource (line 13). Moreover, the sum of
activities” durations, calculated by the rule in line 16, constitutes a (trivial) upper bound on the project
makespan, which is provided by the horizon/1 predicate. Accordingly, the range/1 predicate gathers
all smaller time points at which some activity may still be executed.

The second part of our plain ASP encoding, shown in Listing 3, takes care of generating feasible
schedules of minimal project makespan. For each activity ¢ € V), an answer set includes atoms of
the form undone (¢, 0), undone (i, 1), ..., undone(i,t — 2), undone (¢,t — 1) to represent that ¢ gets
completed at the time point ¢. In turn, activity ¢ is executed at the time points ranging from ¢ — d; to
t — 1, where d; is the duration of 4. Given that ¢ cannot be completed before the time point d;, undone/2
atoms up to d; — 1 are unconditionally derived by the rule in line 20, while the choice rule in line 21
allows for delaying the completion time, as long as the upper bound given by horizon/1 is not exceeded.

Based on atoms undone (¢,t) for a predecessor i of the activity j, the integrity constraint in line 24
requires undone(j,? + d;) to hold, i.e,, j takes at least its duration d; to complete after 7. Ground
instances of this integrity constraint need only be produced up to ¢ + d; matching the upper bound




of horizon/1, since undone(j,t + dj) cannot hold for this threshold. The integrity constraint in
lines 27-28 models resource capacity checks at each time point ¢, where the resource requirements of
all activities ¢ that are neither completed at ¢ nor yet uncompleted at ¢ + d;, i.e., not yet started at ¢,
are added up and compared against the resource capacity. This guarantees that resource capacities are
never exceeded and schedules with the start times .S; = ¢t +1 — d;, for the greatest time point ¢ such that
undone (%,t) belongs to an answer set, are feasible. Finally, makespan minimization is expressed by the
weak constraint in line 31, which penalizes each time point at which some activity without successors
is yet uncompleted by the uniform weight 1. Activities that have successors can be disregarded here
because their completion times will certainly be subsumed by a successor activity executed later on.
The following result establishes the soundness and completeness of our plain ASP encoding of RCPSP.

Theorem 1. Given any RCPSP G = (V, &) along with a schedule {5, };cy, we have that {S;};cyp isa
feasible schedules of minimal project makespan C,.y iff the rules in Listings 2 and 3 yield an optimal
answer set X such that {(7,) | undone(i,t) € X} = {(3,t) | : € V,0 <t < S; + d;} on the fact
representation of G.

Proof. The rules in Listing 2 derive a unique answer set X’ on the fact representation of G, and
X' C X € X'"U{undone(i,t) | i € V,0 <t <) .. d;} holds for any answer set X obtained for
the rules in Listings 2 and 3 on the fact representation of GG. In the following, we consider the set
X = X' U{undone(i,t) | i € V,0 <t < S; +d;} of atoms.

«: Assume that X is an answer set for the rules in Listings 2 and 3 on the fact representation
of G. Then, the integrity constraints in lines 24 and 27-28 check that {.S; };cy satisfies the precedence
and resource constraints, so that {S;}icy is a feasible schedule. Moreover, if X is optimal, for any
schedule {S!};cy such that Cipax > max{S} + d; | i € V}, the fact that X’ U {undone(i,t) | i € V,
0 <t < S/+d;}isnota (better) answer set, which would strictly reduce the penalized ground instances
of the weak constraint in line 31, means that {S}};cy does not satisfy the precedence and resource
constraints. Hence, the optimality of X yields that the project makespan Chax is minimal.

=-: Assume that X is not an optimal answer set for the rules in Listings 2 and 3 on the fact represen-
tation of GG. Then, some of the following cases applies:

1. If Ziev d; < Chax, there is at least one time point smaller than C},,x at which no activity is
executed. That is, Cax cannot be the minimal project makespan.

2. If ZiEV d; > Chax and X does not satisfy some ground instance of the integrity constraint in
line 24, there is an arc (i, j) € £ such that S; < S; + d;. That is, {.S;}icy does not satisfy the
precedence constraints.

3. If ZieV d; > Chax and X does not satisfy some ground instance of the integrity constraint in
lines 27-28, there is a time point ¢ such that ZieA(t) rir. > Ry holds for some resource k € R.
That is, {S; }icy does not satisfy the resource constraints.

4. If neither of the previous cases applies, we have that X is not optimal and there is an (optimal)
answer set Y for which Ciyax > max{t+1 | undone(i,t) € Y'}. By the “<” direction, { S, = t;—
di }iev t;—max{t+1lundone (i, t) ey} i a feasible schedule such that Cyyax > max{S; +d; | i € V}.

O]

A limiting factor regarding the scalability of our plain ASP encoding is its explicit representation
of time points, which get materialized in the grounding phase. However, the availability of #sum
aggregates allows us to represent resource capacity checks at each time point compactly, thanks to a
pseudo-Boolean constraint formulation rather than the clause format of (maximum) satisfiability [53].

3.3. ASP Modulo DL and CASP Encodings

The primary goal of incorporating integer variables as first-class citizens in ASP modulo DL and CASP
is to reduce the ground instantiation size. For RCPSP, we make use of integer variables of the form
(¢,d;) to refer to the completion time of an activity ¢ of duration d; without materializing all candidate

% duration D is lower bound for completing activity I without predecessors
&DIFF{ 0 - (I,D) } <= -D :-task(I,D), first(I).

% completing a successor activity J after I takes at least its duration D’
&DIFF{(I,D) - (J,D’)} <= -D’ :-prec(I,J), task(I,D), task(J,D’).

% case 1 for ignore(I,J): J gets completed before I starts

ignore(I,J) :- check(1,J), task(I,D), task(J,D’), &DIFF{(J,D’) - (I,D)} <= -D.
% case 2 for ignore(I,J): I gets completed before J

ignore(I,J) :-check(1,J), task(I,D), task(J,D’), &DIFF{(I,D) - (J,D’)} <= -1.

% check resource capacity for activity J and all I executed at its completion time
:-res(J,R,X’), capacity(R,C),
#sum{X,I : check(I,J,R), res(I,R,X), not ignore(I,J)} > C-X’.

% completion times of activities without successors provide lower bounds on makespan
&DIFF{(I,D) - makespan} <= O :- task(I,D), final(I).

Listing 4: ASP modulo DL encoding of RCPSP using - -minimize-variable=makespan

values in the grounding phase. Such an integer variable replaces the entire collection of undone (¢,1)
atoms for all time points ¢ in range.

In more detail, the DL constraint in line 20 of Listing 4 states that the duration d; is the smallest
possible completion time of an activity ¢ without predecessors. These lower bounds are carried on
to successor activities j by adding the respective duration d; to the integer variable (i, d;) in line 22,
given that the expression (i,d;) — (j,d;) < —d; is equivalent to (i,d;) +d; < (j,d;).

Lines 24-31 constitute the most sophisticated part of our ASP modulo DL encoding, where resource
capacity checks are expressed without explicitly referencing time points. In the classical Job-shop
Scheduling Problem (JSP) [54, 55, 47], the rule in line 25 (in combination with an integrity constraint
:-check(1,J),not ignore(I,J),not ignore(J,I).) would be sufficient to make sure that the execu-
tion of (distinct) activities 4, j € V sharing a common resource is non-overlapping. In the setting of
RCPSP, however, the activities can be executed simultaneously when the resource capacities are not
exceeded. Hence, the non-overlapping case in line 25, based on the condition (j,d;) — (i,d;) < —d; or
(j,d;) +d; < (i,d;), respectively, may not apply to ¢ and j (and also not to j and 7 in inverse order).
For example, the optimal schedule in Figure 1(b) is such that the execution of activity C' overlaps both
A and F, each of which utilize (1 unit of) the blue resource. Since two units are the capacity limit, the
optimal schedule would be mistakenly discarded if non-overlapping were the sole criterion for capacity
checks. In fact, the schedule is feasible because activity A gets completed before F is started, and the
general observation is that, at each activity’s completion time, at most one other activity requiring
the blue resource is executed simultaneously. We adopt this condition and encode capacity checks at
activities’ completion, serving as events that substitute explicit time points of the plain ASP encoding.

Accordingly, the case that activity j gets completed before ¢ starts is augmented in line 27 by also
deriving ignore (¢, j) when ¢ gets completed before j, as expressed by the condition (7,d;) — (j,d;) <
—lor (i,d;) +1 < (j,dj), respectively. Putting these two conditions together, ignore (¢, j) holds iff
an activity ¢ that shares some resource with j is not currently executed at the completion time of j.
The integrity constraint in lines 30-31 takes the remaining capacity, i.e., the capacity limit reduced by
J’s requirement, as threshold for other activities ¢ executed simultaneously to the completion of j. This
provides an alternative formulation, compared to the plain ASP encoding, of resource capacity checks.

The final constituent is makespan minimization by means of an integer variable makespan, whose
value is in line 34 bound from below by the completion times of activities ¢ without successors,
considering that the expression (¢,d;) — makespan < 0 is equivalent to (¢,d;) < makespan. We
again note that the completion times of activities that have successors are subsumed, and the option
--minimize-variable=makespan instructs clingo[DL] to minimize the makespan variable’s value.

The CASP encoding in Listing 5 is formulated in the input language of clingcon, where the rules in




19
20
21
22
23
24
25
26
27
28
29
30
31
32
33
34
35
36
37
38
39
40
41

% duration D is lower bound for completing activity I without predecessors
&sum{D} <= (I,D) :- task(I,D), first(I).

% completing a successor activity J after I takes at least its duration D’
&sum{(1,D); D’} <= (J,D’) :-prec(I,JT), task(I,D), task(J,D’).

% case 1 for ignore(I,J): J gets completed before I starts

ignore(I,J) :- check(1,J), task(I,D), task(J,D’), &Sum{(J,D’); D} <= (I,D).
% case 2 for ignore(I,J): I gets completed before J

ignore(I,J) :- check(I,J), task(I,D), task(J,D’), &sum{(I,D)} < (J,D’).

% check resource capacity for activity J and all I executed at its completion time
:-res(J,R,X’), capacity(R,C),
#sum{X,I : check(I,J,R), res(I,R,X), not ignore(I,J)} > C-X’.

% completion times of activities without successors provide lower bounds on makespan
&sum{(1,D)} <= makespan :- task(I,D), final(I).

% miminize the makespan (variable’s value)
&MINIMIZE {makespan} .

% activities’ completion times and makespan are restricted by horizon
&oom{D. .H} = (I,D) : - task(I,D), horizon(H).
&pom{0. .H} = makespan : - horizon(H).

Listing 5: CASP encoding of RCPSP

lines 19-34 are mere syntactic mappings from our ASP modulo DL encoding, using clingcon’s &sum
constraints in place of &pirr atoms. Likewise, the &minivize statement in line 37 takes the role of
clingo[DL]’s --minimize-variable=makespan option. Given that the horizon/1 predicate provides a
(trivial) upper bound on the project makespan, the rules in lines 40 and 41 additionally define &pom
atoms limiting integer variables for activities’ completion times and makespan to their feasible domains.

Soundness and completeness according to Theorem 1 also hold for our ASP modulo DL and CASP
encodings.

Proposition 1. Theorem 1 remains valid for either Listing 4 or Listing 5 in place of Listing 3 and
integer variable assignments (i,d;) = S; + d; instead of the condition {(7,t) | undone(i,t) € X} =
{(i,t) | i € V,0 <t < S; + d;} on atoms belonging to an optimal answer set X.

Proof. We show that Theorem 1 remains valid when the integrity constraint in lines 27-28 of Listing 3
is replaced by the following rules:

% case 1 for ignore(I,J): J gets completed before I starts

ignore(I,J) :-check(I,J), task(I,D), undone(I,T-1), not undone(J,T-D), D < T.
% case 2 for ignore(I,J): I gets completed before J

ignore(I,J) :-check(I,J), undone(J,T), not undone(I,T).

% check resource capacity for activity J and all I executed at its completion time
:-res(J,R,X’), capacity(R,C),
#sum{X,I : check(I,J,R), res(I,R,X), not ignore(I,J)} > C-X’.

For (distinct) activities ¢, j € ) that are not connected by any directed path in G and share some resource
k € R,ie,ry > 0andrj; > Ohold, ignore (i, j) isderivedift S; > S;+d; or S;+d; > S;+d; applies.
In turn, ignore (%, j) remains underivable iff we have that S; < S; + d; < S; + d;, i.e., the execution
of activity ¢ has started and is not completed before the completion of j. Hence, the integrity constraint,
which matches the one in lines 30-31 of Listing 4 and Listing 5, makes sure that the accumulated
requirements r;;, of such simultaneously executed activities ¢ do not exceed Ry, — 7, at the completion

Figure 2: Visualization of the benchmark instance J3024_8

time of j for shared resources k£ € R. As the integrity constraint in lines 27-28 of Listing 3 guarantees that
resource capacities are never exceeded, any answer set X obtained for the rules in Listings 2 and 3 on the
fact representation of G yields X U{ignore(i,j) | check(i,j) € X,S; > Sj+d; or Sj+d; > S;+d;}
as an answer set when the integrity constraint is replaced by the above rules. Conversely, an answer
set X obtained with the above rules satisfies the integrity constraint in lines 27-28 of Listing 3, as
establishing Zz‘eA(t) rit, < Ry, at time points ¢ = S; +d; — 1, for activities j € )V with rj;, > 0, suffices.

Given these considerations, we conclude that the ASP modulo DL and CASP encodings in Listing 4
and Listing 5 are sound and complete, as they syntactically reformulate Listing 3, with the above rules
replacing the integrity constraint in lines 27-28, in terms of ASP modulo DL or CASP, respectively. []

4. Experiments

We evaluate our RCPSP modeling approaches on well-known benchmark sets [48] comprising 2,040
small- to medium-scale instances, namely, three times 480 instances with 30, 60, and 90 activities and
another 600 instances with 600 activities." For illustration, Figure 2 visualizes an optimal schedule
for one of the RCPSP instances with 30 activities and 4 resources, where colors indicate the required
resources. As a specific observation, activity 11 starts at the time point 11 although no other activity
(which could be a predecessor or share a common resource) is completed at time 11. Hence, the activity
could safely start at time point 10 already, yet without reducing the optimal project makespan. In
preliminary experiments, we also investigated hard constraints to eliminate such unnecessary delays,
yet observed a deterioration of the solving performance due to increasing the ground instantiation size
by about one order of magnitude. Moreover, we note that the makespan 38 matches the (trivial) lower
bound given by a longest sequence of activities according to the precedence constraints, while this easy
optimality condition does not apply to the majority of RCPSP instances, for which optimality proofs
remain hard.

In our experiments, we ran clingo v5.8.0 on the plain ASP encoding as well as a plain ASP formulation
mimicking the ASP modulo DL encoding idea of resource capacity checks at activities’ completion times
(denoted by ASP_DL-like in Table 2), clingo[DL] v1.5.0 on the ASP modulo DL encoding (abbreviated by
DL in Table 2), and clingcon v5.2.1 on the CASP encoding. For comparison, we also ran the state-of-the-
art CP solver OR-Tools v9.15 on an RCPSP model featuring interval variables and global cumulative
constraints.? Except for OR-Tools, where we determined the option linearization_level:0 as most
effective in preliminary experiments, the solvers were run in their default configurations, which means
that all of them perform model-guided optimization [56]. While core-guided optimization strategies are
available as well, they turned out as significantly less effective and are thus only applied within parallel
portfolios (tested in Table 4). All experiments were run on an Intel® Core™ i7-8650U CPU Dell Latitude
5590 machine under Windows 10.1, imposing a timeout of 900 seconds for each instance.

Table 2 compares the solving performance with the five encodings on the small-scale benchmark
set with 30 activities. The second column reports the number of instances for which a proven optimal
schedule has been obtained within the given 900 seconds, the third column expresses the optimality

'The experimental setup and all details needed for reproducing the presented results are available at https://github.com/
prosysscience/RCPSP-ASPOCP2026.

The source code is made available in the official OR-Tools repository, and can be found at https://raw.githubusercontent.com/
google/or-tools/refs/heads/stable/examples/python/rcpsp_sat.py.

Table 2
Comparison of different encodings on the benchmark set with 30 activities

Encoding Optimal ~ Rate Time (s)
ASP 480 100.0 90.6
ASP_DL-like 472 983 96769
DL 475 98.9 6,535.8
CASP 480 100.0 219.5
OR-Tools 480 100.0 527.9

Table 3
Comparison of different encodings on the benchmark sets with 60, 90, and 120 activities

Activities Encoding Optimal Rate Time (s)

ASP 439 91.5 40,373.7
60 CASP 428  89.2 55,383.4
OR-Tools 435  90.6 46,426.2
ASP 406 84.6 70,470.5
90 CASP 393 819 94,649.1
OR-Tools 399  83.1 78,219.4
ASP 289 48.2 282,284.6
120 CASP 271 452 317,096.7
OR-Tools 287  47.8  296,738.5

rate as percentage of the total number of 480 instances solved to optimality, and the fourth column
reports the cumulative solving time in seconds over all instances. We observe that three encodings
achieve a perfect optimality rate of 100%, namely, ASP, CASP, and OR-Tools. Regarding computational
efficiency, ASP is most performant by requiring only 90.6 seconds in total, followed by CASP with
219.5 seconds, and then OR-Tools with 527.9 seconds. The remaining encodings, ASP_DL-like and
DL, yield proven optimal schedules for 472 (98.3%) or 475 (98.9%) instances, respectively. Although
this shows that clingo[DL] is outperformed by clingcon with an equivalent encoding, clingo[DL] is
still significantly ahead of clingo with the ASP_DL-like encoding, which emphasizes that the reduced
problem representation size thanks to integer variables is beneficial when using the same encoding
idea. However, for the medium-scale instances with 60, 90, and 120 activities, we restrict ourselves to
comparing ASP, CASP, and OR-Tools, while dropping the dominated ASP_DL-like and DL encodings.
Across all three of the benchmark sets with 60, 90, and 120 activities compared in Table 3, ASP
consistently achieves the highest optimality rate and the shortest cumulative solving time, establishing
it as the best-performing encoding overall. We also observe that OR-Tools performs second best and
reverses the previous trend of a longer cumulative solving time than CASP on the small-scale instances.
This may indicate a greater constant overhead of running OR-Tools, which becomes less pronounced as
the instance size and hardness grow. In fact, as the instance size grows, the optimality rates decline with
all encodings, reflecting the increased hardness of the larger instances. In preliminary experiments, we
additionally tested whether clingo’s - -trans-ext option, turning #sum aggregates into clause format by
means of the sequential weight counter encoding [57, 58], can make the optimization more effective in
exchange for a larger problem representation size. The effect was clearly negative, as the size overhead
does not bring about significant savings of search efforts, which would be required to reduce the overall
solving time. We thus conclude that the pseudo-Boolean constraints expressed by #sum aggregates
provide adequate means to model resource capacity checks based on explicitly represented time points.
Table 4 compares the performance of clingo with the plain ASP encoding and OR-Tools in two solving
modes: running the solvers’ (default) portfolios with 8 differently configured search threads in parallel,
and the previously used single-threading mode for which the numbers of proven optimal schedules
are reproduced in parentheses. The results show that increasing the number of threads and processors



Table 4
Comparison of multi-threaded ASP solving and OR-Tools on the benchmark sets with 30, 60, 90, and 120 activities

Encoding 30 60 90 120

ASP 480 (480) 438 (439) 407 (406) 293 (289)
OR-Tools 480 (480) 433 (435) 398 (399) 285 (287)

from 1 to 8 yields no significant improvements in the number of instances solved to optimality for
either clingo or OR-Tools, indicating that parallelism has a negligible impact on optimality proofs within
the given 900 seconds wall clock time. That is, clingo has an edge on OR-Tools in both the single- and
the multi-threaded solving mode, confirming that the advantage of our plain ASP encoding over finite
domain constraint formulations is consistent and independent of the allocated computational resources.

5. Conclusion

We have presented and evaluated several modeling approaches for encoding the RCPSP. In contrast
to previous investigations on the classical JSP [47], where OR-Tools™ interval variables and global
constraints [59] give it an advantage on clingo[DL], while the problem representation size is entirely
prohibitive for plain ASP approaches, we find that clingo with our plain ASP encoding benefits from
the explicit representation of time points and most effectively proves optimal schedules for small- to
medium-scale RCPSP instances. Among the compared approaches incorporating DL or finite domain
constraints, OR-Tools comes closest to clingo, and clingcon performs better than clingo[DL] (and clingo)
with a more sophisticated encoding of capacity checks at specific events instead of all time points in range.
Certainly, the time range impacts the scalability of our plain ASP encoding, and clingo gets along well
with the comparably moderate size of RCPSP instances in the used benchmark set [48]. However, #sum
aggregates or pseudo-Boolean constraints, respectively, turn out to be advantageous for representing
resource capacity limits compactly, and going to the clause format of (maximum) satisfiability would
likely introduce more overhead than can be compensated by means of more performant solving methods.

Despite their moderate size, a significant portion of instances in the benchmark set [48] are highly
combinatorial, given that quite many of those with 60, 90, or 120 activities could not be solved to
optimality yet.* While finding some feasible schedule is easy to accomplish for each instance, gradually
improving the makespan and completing the final optimality proof can be extraordinarily hard. We
observed that the solving performance heavily varies between seemingly similar instances of the same
size, taking just a few seconds or virtual infinity with the same problem encoding and solver. Future
work includes a more detailed investigation of instance characteristics and specific properties of optimal
schedules, which may be available in the literature, in order to identify so-called redundant constraints
with the potential to boost optimality proofs. Straightforward constraints, like prohibiting unnecessary
delayed starts of activities, merely increase the problem representation size without bringing about
significant search savings, so that more elaborate concepts are needed.

